twoway mband calculates cross medians and then graphs the cross medians as a line plot.

Median-band plot of y versus x
    twoway mband y x

Overlaid on a scatterplot of the observed data
    twoway scatter y x || mband y x

As above, but reduce the scatterplot markers to half their normal size
    twoway scatter y x, msize(*.5) || mband y x

As above, but specify 20 bands
    twoway scatter y x, msize(*.5) || mband y x, bands(20)

Add the title “My Title”
    twoway scatter y x || mband y x, title("My Title")

As above, but suppress the legend
    twoway scatter y x || mband y x, title("My Title") legend(off)
2   graph twoway mband — Twoway median-band plots

Syntax

    twoway mband  yvar xvar  [if]  [in]  [ ,  options]

<table>
<thead>
<tr>
<th>options</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>bands(##)</strong></td>
<td>number of bands</td>
</tr>
<tr>
<td><strong>cline_options</strong></td>
<td>change look of the line</td>
</tr>
<tr>
<td><strong>axis_choice_options</strong></td>
<td>associate plot with alternative axis</td>
</tr>
<tr>
<td><strong>twoway_options</strong></td>
<td>titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.</td>
</tr>
</tbody>
</table>

All options are rightmost; see [G-4] Concept: repeated options.

Options

- **bands(##)** specifies the number of bands on which the calculation is to be based. The default is \(\max(10, \text{round}(10 \times \log_{10}(N)))\), where \(N\) is the number of observations.

In a median-band plot, the \(x\) axis is divided into \# equal-width intervals and then the median of \(y\) and the median of \(x\) are calculated in each interval. It is these cross medians that \(\text{mband}\) graphs as a line plot.

- **cline_options** specify how the median-band line is rendered and its appearance; see [G-3] cline_options.

- **axis_choice_options** associate the plot with a particular \(y\) or \(x\) axis on the graph; see [G-3] axis_choice_options.

- **twoway_options** are a set of common options supported by all twoway graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over by() groups, and change some advanced settings. See [G-3] twoway_options.

Remarks and examples

Remarks are presented under the following headings:

Typical use
Use with by()
Typical use

Median bands provide a convenient but crude way to show the tendency in the relationship between $y$ and $x$:

```
. use https://www.stata-press.com/data/r16/auto
   (1978 Automobile Data)
. scatter mpg weight, msize(*.5) || mband mpg weight
```

The important part of the above is “mband mpg weight”. On the scatter, we specified msize(*.5) to make the marker symbols half their normal size; see [G-4] size.

Use with by()

`mband` may be used with `by()` (as can all the `twoway` plot commands):

```
. scatter mpg weight, ms(oh) ||
   mband mpg weight ||, by(foreign, total row(1))
```

In the above graph, we specified `ms(oh)` so as to use hollow symbols; see [G-4] symbolstyle.
Also see

[G-2] **graph twoway fpfit** — Twoway fractional-polynomial prediction plots

[G-2] **graph twoway lfit** — Twoway linear prediction plots

[G-2] **graph twoway line** — Twoway line plots

[G-2] **graph twoway mspline** — Twoway median-spline plots

[G-2] **graph twoway qfit** — Twoway quadratic prediction plots